Combining One-Way Frequency Tables

```
/*Retrieving ODS Table names*/
ods trace on;
proc freq data=sashelp.heart;
      table Status DeathCause Sex Chol_Status BP_Status Weight_Status Smoking_Status;
run;
ods trace off;
```

ODS Table names will be displayed in the Log



Log - (Untitled)

```
258  /*Retrieving ODS Table names*/
259  ods trace on;
260  proc freq data=sashelp.heart;
261       table DeathCause;
262  run;


Output Added:
-------------
Name:        OneWayFreqs
Label:       One-Way Frequencies
Template:    Base.Freq.OneWayFreqs
Path:        Freq.Table1.OneWayFreqs
-------------
```

```
/*Creating output dataset*/
ods output OneWayFreqs=OneWayFreqs(drop=F_Status F_DeathCause F_Sex F_Chol_Status
F_BP_Status F_Weight_Status F_Smoking_Status);
proc freq data=sashelp.heart;
      table Status DeathCause Sex Chol_Status BP_Status Weight_Status Smoking_Status /
<options>;
run;
```

MISSING: treats missing values as nonmissing values
MISSPRINT: displays missing value frequencies

Table	Status	Frequency	Percent	CumFrequency	CumPercent	DeathCause	Sex
Table Status	Alive	3218	61.78	3218	61.78		
Table Status	Dead	1991	38.22	5209	100.00		
Table DeathCause		3218	.	.	.		
Table DeathCause		539	27.07	539	27.07	Cancer	
Table DeathCause		378	18.99	917	46.06	Cerebral Vascular Disease	
Table DeathCause		605	30.39	1522	76.44	Coronary Heart Disease	
Table DeathCause		357	17.93	1879	94.37	Other	
Table DeathCause		112	5.63	1991	100.00	Unknown	
Table Sex		2873	55.15	2873	55.15		Female
Table Sex		2336	44.85	5209	100.00		Male

```
/*Restructure OneWayFreqs*/
/*VarValue is only used when numeric values are entered for categorical variables and
labels are assigned*/
data OneWayFreqs (KEEP = VarName VarFmt VarLabel VarValue frequency percent cumfrequency
cumpercent);
      length      VarName $32
                  VarLabel
                  VarFmt $256
                  VarValue 8;
      set OneWayFreqs;
      array vars {*} Status DeathCause Sex Chol_Status BP_Status Weight_Status
Smoking_Status;
            VarName=SCAN(table,2,' ');
            do i = 1 to DIM(vars);
                  if UPCASE(VarName)=UPCASE(VNAME(vars(i))) then do;
                  VarFmt =VVALUE(vars(i));
                  VarLabel =VLABEL(vars(i));
                  VarValue =vars(i);
                  return;
            end;
      end;
run;
```

array: Defines a set of elements that you plan to process as a group.
scan: Returns the nth word from a character string
VNAME: Assigns a variable name as the value of a specified variable.
VVALUE: Returns the formatted value of a specified variable.
VLABEL: Returns the label of a specified variable.

VarName	VarLabel	VarFmt	VarValue	Frequency	Percent	CumFrequency	CumPercent
Status	Status	Alive	.	3218	61.78	3218	61.78
Status	Status	Dead	.	1991	38.22	5209	100.00
DeathCause	Cause of Death		.	3218	.	.	.
DeathCause	Cause of Death	Cancer	.	539	27.07	539	27.07
DeathCause	Cause of Death	Cerebral Vascular Disease	.	378	18.99	917	46.06
DeathCause	Cause of Death	Coronary Heart Disease	.	605	30.39	1522	76.44
DeathCause	Cause of Death	Other	.	357	17.93	1879	94.37
DeathCause	Cause of Death	Unknown	.	112	5.63	1991	100.00
Sex	Sex	Female	.	2873	55.15	2873	55.15
Sex	Sex	Male	.	2336	44.85	5209	100.00

```
/*Adding labels*/
data OneWayFreqs;
      set OneWayFreqs;
      LABEL       VarName="Variable Name"
                  VarLabel="Variable"
                  VarFmt="Categories"
                  VarValue="Value";
      if VarFmt="" then VarFmt="Missing";
run;

/*Print Table*/
proc print data=OneWayFreqs noobs label;
      var VarLabel VarFmt frequency percent;
run;
```

noobs: Suppresses the observation number in the output
label: Uses variables' labels as column headings

Variable	Categories	Frequency	Percent
Status	Alive	3218	61.78
Status	Dead	1991	38.22
Cause of Death	Missing	3218	.
Cause of Death	Cancer	539	27.07
Cause of Death	Cerebral Vascular Disease	378	18.99
Cause of Death	Coronary Heart Disease	605	30.39
Cause of Death	Other	357	17.93
Cause of Death	Unknown	112	5.63
Sex	Female	2873	55.15
Sex	Male	2336	44.85

Restructuring PROC MEANS

```
/*Output results to a dataset*/
proc means data=sashelp.heart maxdec=2;
      var AgeCHDdiag AgeAtStart Height Weight Diastolic Systolic MRW AgeAtDeath
Cholesterol;
      output out=var_means;
run;
ods output summary=var_means; DOES NOT WORK
```

Obs	_TYPE_	_FREQ_	_STAT_	AgeCHDdiag	AgeAtStart	Height	Weight	Diastolic	Systolic	MRW	AgeAtDeath	Cholesterol
1	0	5209	N	1449.00	5209.00	5203.00	5203.00	5209.00	5209.00	5203.00	1991.00	5057.00
2	0	5209	MIN	32.00	28.00	51.50	67.00	50.00	82.00	67.00	36.00	96.00
3	0	5209	MAX	90.00	62.00	76.50	300.00	160.00	300.00	268.00	93.00	568.00
4	0	5209	MEAN	63.30	44.07	64.81	153.09	85.36	136.91	119.96	70.54	227.42
5	0	5209	STD	10.02	8.57	3.58	28.92	12.97	23.74	19.98	10.56	44.94

```
/*Dropping _TYPE_ and _FREQ_*/
proc means data=sashelp.heart maxdec=2;
      var AgeCHDdiag AgeAtStart Height Weight Diastolic Systolic MRW AgeAtDeath
Cholesterol;
      output out=var_means(drop=_FREQ_ _TYPE_);
run;

/*Transpose dataset*/
proc transpose data=var_means out=trans_mean(rename=(_NAME_=VarName));
      id _STAT_;
run;
```

VarName	VarLabel	N	MIN	MAX	MEAN	STD
AgeCHDdiag	Age CHD Diagnosed	1449	32.0	90.0	63.303	10.0182
AgeAtStart	Age at Start	5209	28.0	62.0	44.069	8.5750
Height		5203	51.5	76.5	64.813	3.5827
Weight		5203	67.0	300.0	153.087	28.9154

Adding information from PROC UNIVARIATE

```
/*Creating output datasets*/
ods output BasicMeasures=BasicMeasures;
proc univariate data=sashelp.heart;
      var AgeCHDdiag AgeAtStart Height Weight Diastolic Systolic MRW AgeAtDeath
Cholesterol;
run;

proc sort data=BasicMeasures; by VarName; run;

/*Restructuring BasicMeasures to create a composite summary dataset with one row per
variable*/
data BasicMeasures (keep = VarName Median);
      length VarName $32;
      set BasicMeasures;
      by VarName;
      if LocMeasure="Median" then Median=LocValue;
      if LAST.VarName then
      output BasicMeasures;
      retain Median;
run;

proc sort data=trans_mean; by VarName; run;

/*Create final composite dataset*/
data UnivariateStats;
      length VarLabel $256;
      merge  Trans_Mean BasicMeasures ;
      by VarName;
run;

/*Adding Labels*/
data UnivariateStats;
      set UnivariateStats;
      LABEL       VarName="Variable Name"
                  VarLabel="Variable"
                  MIN="Min"
                  MAX="Max"
                  MEAN="Mean"
                  STD="Standard Deviation";
      if VarLabel="" then VarLabel=VarName;
run;.

/*Print Table*/
proc print data=UnivariateStats noobs label;
      var VarLabel n mean std median;
run;
```

Variable	N	Mean	Standard Deviation	Median
Age at Death	1991	70.536	10.5594	71.0
Age at Start	5209	44.069	8.5750	43.0
Age CHD Diagnosed	1449	63.303	10.0182	63.0
Cholesterol	5057	227.417	44.9355	223.0
Diastolic	5209	85.359	12.9731	84.0
Height	5203	64.813	3.5827	64.5
Metropolitan Relative Weight	5203	119.958	19.9834	118.0
Systolic	5209	136.910	23.7396	132.0
Weight	5203	153.087	28.9154	150.0